
1.1 Change 1
	Testcase name
	8.1.2.3

	Protocol
	RRC

	Function Name
	f_TC_8_1_2_3_EUTRA()

	Reason for change
	1. Since UE triggered establishment of a dedicated EPS bearer context is a UE optional feature and support is indicated by parameter pc_ESM_MO_Bearer_Allocation defined in 36.523-2 Table A.4.4-1/18, hence to avoid dependency on optional feature, the UE is made to attempt an outgoing call by paging the UE.

2. As per the note mentioned in step 3 of the prose of this CR, the SS should be configured to not resolve the contention. Hence the SS should be configured to not send the contention resolution message prior to ignoring the rrcConnectionRequest at step 2 and then reset the default settings after the T300 timer expiry prior to step 4.  
3. Since the UE is being Paged, the establishment cause for rrcConnectionRequest message should be changed to MT-Access. 
4. As per the notes mentioned in step 3 of the prose and due to the change mentioned above in step 2, UE would retry the RACH procedure and hence send multiple rrcConnectionRequest. SS should handle and ignore the multiple RRC connection request attempts
Note : Prose CR will be submitted in the next RAN5 meeting

	Summary of change
	1. At step 1, instead of an AT command to trigger dedicated radio bearer establishment, a Paging message is sent to the UE. 
2. SS configured to not send the contention resolution message to UE prior to step 2 and reverted back with the default setting of sending contention resolution prior to step 4.
3. At step 2, establishment cause of rrcConnectionRequest changed to MT-Access.  
4. At step 3, multiple rrcConnectionRequest attempts now being handled and ignored  in TTCN

	Source of change
	RRC_ConnEst.ttcn


Before:
	  function f_TC_8_1_2_3_EUTRA() runs on EUTRA_PTC

  { /* RRC Connection Establishment in RRC_IDLE state: return to idle state after T300 timeout */

    f_EUTRA_Init ( c1 );

    //Create and configure cell

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_TestBody_Set ( true );

    //@siclog "Step 1" siclog@

    //Make the UE attempt an outgoing call

    f_UT_RequestActivateDedicatedBearer (UT, hex2int(tsc_EpsDedicatedBearerId), hex2int(tsc_EpsDefaultBearerId), cs_508_EPS_QoS_Dedicated_1, cs_508_Tft_Dedicated_1);
    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionRequest message.

    SRB.receive( car_SRB0_RrcPdu_IND( eutra_Cell1,

                                      cr_RRCConnectionRequest ( cr_EstablishmentCause_MoData ) ) );

    //@siclog "Step 3" siclog@

    //The SS waits for 2s. Note: the UE may transmit one or more RRCConnectionRequest messages but the SS does not answer to these messages.

    f_Delay ( f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, rrcTimer, 2.0 ) ); //@sic R5w100302 sic@
    //@siclog "Step 4" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.2 indicate that the UE is in E-UTRA RRC_IDLE state on Cell 1?

    f_EUTRA_508Check_IdleState ( eutra_Cell1 );

    f_EUTRA_TestBody_Set ( false );

    /* Postamble */

    f_EUTRA_Postamble ( eutra_Cell1, E1_IDLE );

  }


After:
	import from EUTRA_CellCfg_Templates all;
 template (value) RachProcedureConfig_Type cs_RachProcedureConfig_8123 (RACH_TimingAdvance_Type p_RAR_TA,

                                                                          Dl_Bandwidth_Type p_Dl_Bandwidth):=

  { /* Rach Response config for test case 7.1.2.3 */

    /* @status    APPROVED */

    RACH_ConfigCommon := omit,

    RACH_ConfigDedicated := omit,

    RachProcedureList := {

      // preamble reception 3.  RAR Tx succesful but no Contention Resolution

      {

        RAResponse := cs_RandomAccessResponseConfig_Def(p_RAR_TA,p_Dl_Bandwidth),

        ContentionResolutionCtrl := cs_ContReslCtrl_None_TCRNTI

      }

    }

  };
  function f_TC_8_1_2_3_EUTRA() runs on EUTRA_PTC

  { /* RRC Connection Establishment in RRC_IDLE state: return to idle state after T300 timeout */

    timer t_Watchdog; //general purpose timer . 

    var Dl_Bandwidth_Type v_DlBandwidth := n25;  /* for 5 MHz; @sic R5s100073, R5s100291: add. change sic@ */
    f_EUTRA_Init ( c1 );

    v_DlBandwidth := f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1);  
    //Create and configure cell

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */

    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);

    f_EUTRA_TestBody_Set ( true );

    //@siclog "Step 1" siclog@
    f_EUTRA_SS_ConfigRachProcedure (eutra_Cell1, omit, cs_RachProcedureConfig_8123
    (f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1), v_DlBandwidth));
    f_EUTRA_UE_Page_Def ( eutra_Cell1 );
    //@siclog "Step 2" siclog@

    //The UE transmits an RRCConnectionRequest message.

    SRB.receive( car_SRB0_RrcPdu_IND( eutra_Cell1,

                                      cr_RRCConnectionRequest ( cr_EstablishmentCause_mt_Access) ) );

    //@siclog "Step 3" siclog@

    //The SS waits for 2s. Note: the UE may transmit one or more RRCConnectionRequest messages but the SS does not answer to these messages.

    t_Watchdog.start ( f_EUTRA_SetTimerToleranceMax ( eutra_Cell1, rrcTimer, 2.0 ) ); //@sic R5w100302 sic@

    alt {

      [] t_Watchdog.timeout {}

      [] SRB.receive( car_SRB0_RrcPdu_IND( eutra_Cell1,

                                      cr_RRCConnectionRequest ( cr_EstablishmentCause_mt_Access) ) )

      {

        repeat;

      }

    }

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, omit, cs_RachProcedureConfig_Def(omit, f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1),

                                                                                       v_DlBandwidth ));
    //@siclog "Step 4" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.2 indicate that the UE is in E-UTRA RRC_IDLE state on Cell 1?

    f_EUTRA_508Check_IdleState ( eutra_Cell1 );

    f_EUTRA_TestBody_Set ( false );

    /* Postamble */

    f_EUTRA_Postamble ( eutra_Cell1, E1_IDLE );

  }

